Appendix

```
/* For Stata versions 11 and later */
* Logistic regression adjusting for confounders
* Outcome is "bmi3" – overweight/obese BMI at wave 3
* Exposure is low physical activity (inactive = 1)
/* Method 1 – Marginal standardization (standard = TOTAL population) */
      * First run logistic regression
      logit bmi3 i.inactive age18 female white diet bmi2, or
      * Estimate prevalence difference for inactive = 1 vs. inactive = 0
      margins inactive, contrast(effects)
      * Alternative syntax
      margins inactive, dydx(inactive)
      * Predicted probabilities by exposure level
      margins inactive, post
      * Estimate prevalence ratio for inactive = 1 vs. inactive = 0
      margins, coeflegend
      nlcom (ratio1: _b[1.inactive]/_b[0bn.inactive]), post
      test ratio 1 = 1
             /* Gives appropriate p value for comparing ratio1 to 1 */
   /* Method 1 – Marginal standardization (standard = EXPOSED population) */
      * First run logistic regression
      logit bmi3 i.inactive age18 female white diet bmi2, or
      * Estimate prevalence difference for inactive = 1 vs. inactive = 0
      margins if inactive==1, dydx(inactive)
      * Predicted probabilities by exposure level
      margins inactive, subpop(inactive) post
             /* subpop option calculates weights based on inactive = 1 (exposed group) */
      * Estimate prevalence ratio for inactive = 1 vs. inactive = 0
      margins, coeflegend
      nlcom (ratio1: b[1.inactive]/ b[0bn.inactive]), post
      test ratio 1 = 1
            /* Gives appropriate p value for comparing ratio1 to 1 */
```

```
************
/* Method 2 – Prediction at the Modes */
/* Note: User must calculate the mode for each confounder */
      logit bmi3 i.inactive age18 female white diet bmi2, or
      margins inactive, at(age18=1 female=1 white=1 diet=0 bmi2=0) contrast(effects)
      margins inactive, at(age18=1 female=1 white=1 diet=0 bmi2=0) post
      margins, coeflegend
      nlcom (ratio1: _b[1.inactive]/_b[0bn.inactive]), post
      test ratio 1 = 1
*********
/* Method 3 – Prediction at the Means */
      logit bmi3 i.inactive age18 female white diet bmi2, or
      * atmeans option sets each confounder to its mean value
      margins inactive, atmeans contrast(effects)
      margins inactive, atmeans post
      margins, coeflegend
      nlcom (ratio1: _b[1.inactive]/_b[0bn.inactive]), post
      test ratio1 = 1
```